

	Math	Simple Lisp
Addition	$a + b + c$	<code>(sum a b c)</code>
Subtraction	$a - b$	<code>(sum a (product -1 b))</code>
Multiplication	abc	<code>(product a b c)</code>
Division	a/b	<code>(product a (power b -1))</code>
Power	a^b	<code>(power a b)</code>
Derivative	$\partial f / \partial x$	<code>(derivative f x)</code>
Component	A^1_2	<code>(product A12 (tensor 1 2))</code>

Symbolic expressions

Products of sums are expanded.

```
? (product a (sum b c))
(sum (product a b) (product a c))

? (power (sum a b) 2)
(sum (power a 2) (power b 2) (product 2 a b))
```

Sums in an exponent are expanded.

```
? (power a (sum b c))
(product (power a b) (power a c))
```

Vectors, matrices, and tensors are written as sums of components.

The following example computes the inner product of two vectors A and B .

$$A = \begin{pmatrix} A_1 \\ A_2 \end{pmatrix}, \quad B = \begin{pmatrix} B_1 \\ B_2 \end{pmatrix}, \quad A \cdot B = A_1 B_1 + A_2 B_2$$

```
? (setq A (sum (product A1 (tensor 1)) (product A2 (tensor 2))))
? (setq B (sum (product B1 (tensor 1)) (product B2 (tensor 2))))
? (dot A B)
(sum (product A1 B1) (product A2 B2))
```

Tensor components can use symbolic indices. The following example is the same as above except x and y are used for the index names.

```
? (setq A (sum (product A1 (tensor x)) (product A2 (tensor y))))
? (setq B (sum (product B1 (tensor x)) (product B2 (tensor y))))
? (dot A B)
(sum (product A1 B1) (product A2 B2))
```

GR example

Define the metric tensor.

$$g_{\mu\nu} = \begin{pmatrix} -\xi(r) & 0 & 0 & 0 \\ 0 & 1/\xi(r) & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \sin^2 \theta \end{pmatrix}$$

```
(setq gtt (product -1 (xi r)))
(setq grr (power (xi r) -1))
(setq gthetatheta (power r 2))
(setq gphiphi (product (power r 2) (power (sin theta) 2)))
```

```
(setq gdd (sum
  (product gtt (tensor t t))
  (product grr (tensor r r))
  (product gthetatheta (tensor theta theta))
  (product gphiphi (tensor phi phi))
))
```

Calculate $g^{\mu\nu} = (g_{\mu\nu})^{-1}$.

```
(setq g (determinant gdd t r theta phi))
(setq guu (product (power g -1) (adjunct gdd t r theta phi)))
```

Calculate connection coefficients.

$$\Gamma^\alpha_{\beta\gamma} = \frac{1}{2} g^{\alpha\mu} (g_{\mu\beta,\gamma} + g_{\mu\gamma,\beta} - g_{\beta\gamma,\mu})$$

```
(define gradient (sum
  (product (derivative arg t) (tensor t))
  (product (derivative arg r) (tensor r))
  (product (derivative arg theta) (tensor theta))
  (product (derivative arg phi) (tensor phi))
))
```

```
(setq gddd (gradient gdd))
```

```
(setq GAMDDD (product 1/2 (sum
  gddd
  (transpose 2 3 grad)
  (product -1 (transpose 1 2 (transpose 2 3 grad))) ; transpose bc,a to (,a)bc
)))
```

```
(setq GAMUDD (contract 2 3 (product guu GAMDDD)))
```

Calculate Riemann tensor.

$$R^\alpha_{\beta\gamma\delta} = \Gamma^\alpha_{\beta\delta,\gamma} - \Gamma^\alpha_{\beta\gamma,\delta} + \Gamma^\alpha_{\mu\gamma} \Gamma^\mu_{\beta\delta} - \Gamma^\alpha_{\mu\delta} \Gamma^\mu_{\beta\gamma}$$

```
(setq GAMUDDD (gradient GAMUDD))
```

```
(setq GAMGAM (contract 2 4 (product GAMUDD GAMUDD)))
```

```
(setq RUDDD (sum  
  (transpose 3 4 GAMUDDD)  
  (product -1 GAMUDDD)  
  (transpose 2 3 GAMGAM)  
  (product -1 (transpose 3 4 (transpose 2 3 GAMGAM))))  
)
```

Calculate Ricci tensor.

$$R_{\mu\nu} = R^\alpha_{\mu\alpha\nu}$$

```
(setq RDD (contract 1 3 RUDDD))
```

Calculate Ricci scalar.

$$R = R^\mu_{\mu}$$

```
(setq R (contract 1 2 (contract 2 3 (product guu RDD))))
```

Calculate Einstein tensor.

$$G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2}g_{\mu\nu}R$$

```
(setq GDD (sum RDD (product -1/2 gdd R)))
```