Two spins

The spin state $|s\rangle$ for two spins is a unit vector in $\mathbb{C}^4$.

$$|s\rangle = \begin{pmatrix} c_{++} \\ c_{+-} \\ c_{-+} \\ c_{--} \end{pmatrix}, \quad |c_{++}|^2 + |c_{+-}|^2 + |c_{-+}|^2 + |c_{--}|^2 = 1$$

Spin measurement probabilities are the transition probabilities from $|s\rangle$ to an eigenstate.

For spin measurements in the z direction we have

$$\begin{aligned}
\Pr\left(S_{1z} = +\tfrac{\hbar}{2} \text{ and } S_{2z} = +\tfrac{\hbar}{2}\right) &= |\langle z_{++}|s\rangle|^2 = |c_{++}|^2 \\
\Pr\left(S_{1z} = +\tfrac{\hbar}{2} \text{ and } S_{2z} = -\tfrac{\hbar}{2}\right) &= |\langle z_{+-}|s\rangle|^2 = |c_{+-}|^2 \\
\Pr\left(S_{1z} = -\tfrac{\hbar}{2} \text{ and } S_{2z} = +\tfrac{\hbar}{2}\right) &= |\langle z_{-+}|s\rangle|^2 = |c_{-+}|^2 \\
\Pr\left(S_{1z} = -\tfrac{\hbar}{2} \text{ and } S_{2z} = -\tfrac{\hbar}{2}\right) &= |\langle z_{--}|s\rangle|^2 = |c_{--}|^2
\end{aligned}$$

where the eigenstates are

$$z_{++} = \begin{pmatrix} 1 \\ 0 \\ 0 \\ 0 \end{pmatrix}, \quad z_{+-} = \begin{pmatrix} 0 \\ 1 \\ 0 \\ 0 \end{pmatrix}, \quad z_{-+} = \begin{pmatrix} 0 \\ 0 \\ 1 \\ 0 \end{pmatrix}, \quad z_{--} = \begin{pmatrix} 0 \\ 0 \\ 0 \\ 1 \end{pmatrix}$$

Spin operators for the first spin ($\otimes$ is kronecker product).

$$\begin{aligned}
S_{1x} &= \frac{\hbar}{2}\sigma_x \otimes I = \frac{\hbar}{2}\begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix} \\
S_{1y} &= \frac{\hbar}{2}\sigma_y \otimes I = \frac{\hbar}{2}\begin{pmatrix} 0 & 0 & -i & 0 \\ 0 & 0 & 0 & -i \\ i & 0 & 0 & 0 \\ 0 & i & 0 & 0 \end{pmatrix} \\
S_{1z} &= \frac{\hbar}{2}\sigma_z \otimes I = \frac{\hbar}{2}\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}
\end{aligned}$$

Spin operators for the second spin.

$$S_{2x} = \frac{\hbar}{2} I \otimes \sigma_x = \frac{\hbar}{2} \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

$$S_{2y} = \frac{\hbar}{2} I \otimes \sigma_y = \frac{\hbar}{2} \begin{pmatrix} 0 & -i & 0 & 0 \\ i & 0 & 0 & 0 \\ 0 & 0 & 0 & -i \\ 0 & 0 & i & 0 \end{pmatrix}$$

$$S_{2z} = \frac{\hbar}{2} I \otimes \sigma_z = \frac{\hbar}{2} \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

Expectation values for the first spin.

$$\begin{aligned} \langle S_{1x} \rangle &= \langle s | S_{1x} | s \rangle = \tfrac{\hbar}{2} \left(c_{++} c_{-+}^* + c_{++}^* c_{-+} + c_{+-} c_{--}^* + c_{+-}^* c_{--} \right) \\ \langle S_{1y} \rangle &= \langle s | S_{1y} | s \rangle = \tfrac{i\hbar}{2} \left(c_{++} c_{-+}^* - c_{++}^* c_{-+} + c_{+-} c_{--}^* - c_{+-}^* c_{--} \right) \\ \langle S_{1z} \rangle &= \langle s | S_{1z} | s \rangle = \tfrac{\hbar}{2} \left(|c_{++}|^2 + |c_{+-}|^2 - |c_{-+}|^2 - |c_{--}|^2 \right) \end{aligned}$$

Expectation values for the second spin.

$$\begin{aligned} \langle S_{2x} \rangle &= \langle s | S_{2x} | s \rangle = \tfrac{\hbar}{2} \left(c_{++} c_{+-}^* + c_{++}^* c_{+-} + c_{-+} c_{--}^* + c_{-+}^* c_{--} \right) \\ \langle S_{2y} \rangle &= \langle s | S_{2y} | s \rangle = \tfrac{i\hbar}{2} \left(c_{++} c_{+-}^* - c_{++}^* c_{+-} + c_{-+} c_{--}^* - c_{-+}^* c_{--} \right) \\ \langle S_{2z} \rangle &= \langle s | S_{2z} | s \rangle = \tfrac{\hbar}{2} \left(|c_{++}|^2 - |c_{+-}|^2 + |c_{-+}|^2 - |c_{--}|^2 \right) \end{aligned}$$

Let

$$\mathbf{S}_1 = \begin{pmatrix} S_{1x} \\ S_{1y} \\ S_{1z} \end{pmatrix}, \quad \mathbf{S}_2 = \begin{pmatrix} S_{2x} \\ S_{2y} \\ S_{2z} \end{pmatrix}$$

Total spin is given by

$$(\mathbf{S}_1 + \mathbf{S}_2)^2 |s\rangle = \hbar^2 \begin{pmatrix} 2c_{++} \\ c_{+-} + c_{-+} \\ c_{+-} + c_{-+} \\ 2c_{--} \end{pmatrix}$$

Expectation value for total spin.

$$\langle s | (\mathbf{S}_1 + \mathbf{S}_2)^2 | s \rangle = \hbar^2 \left(2|c_{++}|^2 + |c_{+-} + c_{-+}|^2 + 2|c_{--}|^2 \right)$$

1. Verify spin operators for two spins.

```
sigmax = ((0,1),(1,0))
sigmay = ((0,-i),(i,0))
sigmaz = ((1,0),(0,-1))

I = ((1,0),(0,1))

S1x = 1/2 hbar kronecker(sigmax,I)
S1y = 1/2 hbar kronecker(sigmay,I)
S1z = 1/2 hbar kronecker(sigmaz,I)

S2x = 1/2 hbar kronecker(I,sigmax)
S2y = 1/2 hbar kronecker(I,sigmay)
S2z = 1/2 hbar kronecker(I,sigmaz)

check(S1x == 1/2 hbar ((0,0,1,0),(0,0,0,1),(1,0,0,0),(0,1,0,0)))
check(S1y == 1/2 hbar ((0,0,-i,0),(0,0,0,-i),(i,0,0,0),(0,i,0,0)))
check(S1z == 1/2 hbar ((1,0,0,0),(0,1,0,0),(0,0,-1,0),(0,0,0,-1)))

check(S2x == 1/2 hbar ((0,1,0,0),(1,0,0,0),(0,0,0,1),(0,0,1,0)))
check(S2y == 1/2 hbar ((0,-i,0,0),(i,0,0,0),(0,0,0,-i),(0,0,i,0)))
check(S2z == 1/2 hbar ((1,0,0,0),(0,-1,0,0),(0,0,1,0),(0,0,0,-1)))
```

2. Verify expectation values for two spins.

```
S1x = 1/2 hbar ((0,0,1,0),(0,0,0,1),(1,0,0,0),(0,1,0,0))
S1y = 1/2 hbar ((0,0,-i,0),(0,0,0,-i),(i,0,0,0),(0,i,0,0))
S1z = 1/2 hbar ((1,0,0,0),(0,1,0,0),(0,0,-1,0),(0,0,0,-1))

S2x = 1/2 hbar ((0,1,0,0),(1,0,0,0),(0,0,0,1),(0,0,1,0))
S2y = 1/2 hbar ((0,-i,0,0),(i,0,0,0),(0,0,0,-i),(0,0,i,0))
S2z = 1/2 hbar ((1,0,0,0),(0,-1,0,0),(0,0,1,0),(0,0,0,-1))

cpp = xpp + i ypp
cpm = xpm + i ypm
cmp = xmp + i ymp
cmm = xmm + i ymm

s = (cpp,cpm,cmp,cmm)

check(dot(conj(s),S1x,s) ==
1/2 hbar (cpp conj(cmp) + conj(cpp) cmp + cpm conj(cmm) + conj(cpm) cmm))

check(dot(conj(s),S1y,s) ==
1/2 i hbar (cpp conj(cmp) - conj(cpp) cmp + cpm conj(cmm) - conj(cpm) cmm))

check(dot(conj(s),S1z,s) ==
1/2 hbar (cpp conj(cpp) + cpm conj(cpm) - cmp conj(cmp) - cmm conj(cmm)))

check(dot(conj(s),S2x,s) ==
1/2 hbar (cpp conj(cpm) + conj(cpp) cpm + cmp conj(cmm) + conj(cmp) cmm))

check(dot(conj(s),S2y,s) ==
1/2 i hbar (cpp conj(cpm) - conj(cpp) cpm + cmp conj(cmm) - conj(cmp) cmm))

check(dot(conj(s),S2z,s) ==
1/2 hbar (cpp conj(cpp) - cpm conj(cpm) + cmp conj(cmp) - cmm conj(cmm)))
```