# Eigenmath Manual

# October 29, 2022

# Contents

1	$\mathbf{Intr}$	roduction	3		
	1.1	Syntax	. 3		
	1.2	Testing for equality			
	1.3	Arithmetic	. 5		
	1.4	Exponents	. 6		
	1.5	Symbols	. 7		
	1.6	Functions	. 8		
	1.7	Complex numbers	. 9		
	1.8	Linear algebra	. 11		
	1.9	Component arithmetic	. 12		
2	Calculus 13				
	2.1	Derivative	. 13		
	2.2	Gradient	. 13		
	2.3	Template functions	. 14		
	2.4	Integral	. 15		
	2.5	Arc length	. 16		
	2.6	Line integrals	. 18		
	2.7	Surface area	. 20		
	2.8	Surface integrals	. 21		
	2.9	Green's theorem	. 22		
	2.10	Stokes's theorem	. 24		
3	Qua	antum Computing	26		
4	Dra	w (macOS)	30		
5	Fun	ction Reference	33		
6	Tric	:ks	56		

Eigenmath is available for both macOS and Linux.

For macOS, commands are entered in the following field.



Multiple commands can be put together in a script. Scripts are run by clicking the Run button.



After a script runs, all of the results are available in command mode.

### 1 Introduction

Consider the following arithmetic from Vladimir Nabokov's autobiography "Speak, Memory."

A foolish tutor had explained logarithms to me much too early, and I had read (in a British publication, the *Boy's Own Paper*, I believe) about a certain Hindu calculator who in exactly two seconds could find the seventeenth root of, say, 3529471145760275132 301897342055866171392 (I am not sure I have got this right; anyway the root was 212).

Let us compute 212<sup>17</sup> and check the result. At the Eigenmath prompt, enter

```
212^17
```

The result is

3529471145760275132301897342055866171392

Now let us compute the seventeenth root of this number.

```
N = 212^17
N N = 3529471145760275132301897342055866171392 N^(1/17) 212
```

# 1.1 Syntax

Arithmetic operators have the expected precedence of multiplication and division before addition and subtraction. Subexpressions in parentheses have highest precedence.

Math	Eigen math	Comment
a = b	a == b	test for equality
-a	-a	negation
a + b	a+b	addition
a-b	a-b	subtraction
ab	a b	$multiplication,\ also\ {\tt a*b}$
$rac{a}{b}$	a/b	division
$\frac{a}{bc}$	a/b/c	division is left-associative
$a^2$	a^2	power
$\sqrt{a}$	sqrt(a)	square root, also a^(1/2)
a(b+c)	a (b+c)	space is required
f(a)	f(a)	function
$\begin{pmatrix} a \\ b \\ c \end{pmatrix}$	(a,b,c)	vector
$\begin{pmatrix} a & b \\ c & d \end{pmatrix}$	((a,b),(c,d))	matrix
$F^1{}_2$	F[1,2]	tensor component access
	"hello, world"	string literal
$\pi$	pi	
e	exp(1)	natural number

## 1.2 Testing for equality

The infix operator == is used to test for equality of operands. The operator evaluates to 1 if the operands are equal and 0 if the operands are not equal.

```
exp(i pi) == -1
```

The == operator uses simplify internally. In effect, A==B is equivalent to simplify(A-B)==0.

#### 1.3 Arithmetic

Integers and rational numbers can have any number of digits, regardless of native word size. For example,  $212^{17} \gg 64$  bits.

#### 2^64

18446744073709551616

#### 212^17

3529471145760275132301897342055866171392

Integer and rational number arithmetic is used by default.

```
1/2 + 1/3
<sub>5</sub>
```

Floating point arithmetic can also be used.

```
1/2 + 1/3.0
```

0.833333

An integer or rational number result can be converted to a floating point value by entering float.

#### 212^17

#### 3529471145760275132301897342055866171392

#### float

```
3.52947 \times 10^{39}
```

The following example shows how to enter a floating point value using scientific notation.

```
epsilon = 1.0 10^(-6) epsilon \varepsilon = 1.0 \times 10^{-6}
```

### 1.4 Exponents

Parentheses are required around negative exponents. For example,

```
10^(-3)
```

instead of

#### 10^-3

The reason for this is that the binding of the negative sign is not always obvious. For example, consider

#### $x^-1/2$

It is not clear whether the exponent should be -1 or -1/2. Hence the following syntax is required.

```
x^{(-1/2)}
```

In general, parentheses are always required when the exponent is an expression. For example,  $x^1/2$  is evaluated as  $(x^1)/2$  which is probably not the desired result.

```
x^1/2
\frac{1}{2}x
Using x^(1/2) yields the desired result.
x^{1/2}
```

## 1.5 Symbols

Symbols are defined using an equals sign.

```
N = 212^17
```

No result is printed when a symbol is defined. To see the value of a symbol, just evaluate it.

N

```
N = 3529471145760275132301897342055866171392
```

Symbols can have more that one letter. Everything after the first letter is displayed as a subscript.

```
NA = 6.02214 10<sup>23</sup>
```

$$N_A = 6.02214 \times 10^{23}$$

A symbol can be the name of a Greek letter.

```
xi = 1/2
xi
\xi = \frac{1}{2}
```

Greek letters can appear in subscripts.

```
\begin{array}{l} {\rm Amu} \ = \ 2.0 \\ {\rm Amu} \\ \\ A_{\mu} = 2.0 \end{array}
```

The following example shows how a symbol is scanned to find Greek letters.

```
alphamunu = 1
alphamunu
```

```
\alpha_{\mu\nu} = 1
```

Symbol definitions are evaluated serially until a terminal symbol is reached. The following example sets A = B followed by B = C. Then when A is evaluated, the result is C.

```
A = B
B = C
A
A = C
```

Although A = C is printed, inside the program the binding of A is still B, as can be seen with the binding function.

```
binding(A)
```

B

The quote function returns its argument unevaluated and can be used to clear a symbol. The following example clears A so that its evaluation goes back to being A instead of C.

```
A = quote(A)
A
A
```

#### 1.6 Functions

The syntax for defining functions is function-name (arg-list) = expr where arg-list is a comma separated list of zero to nine symbols that receive arguments. Unlike symbol definitions, expr is not evaluated when function-name is defined. Instead, expr is evaluated when function-name is used in a subsequent computation.

The following example defines a sinc function and evaluates it at  $\pi/2$ .

```
f(x) = \sin(x)/x
f(pi/2)
```

```
\frac{2}{\pi}
```

The scope of function arguments is the function definition.

After a user function is defined, *expr* can be recalled using the binding function.

## binding(f)

```
\frac{\sin(x)}{x}
```

# 1.7 Complex numbers

Symbol i is initialized to  $\sqrt{-1}$ .

Complex quantities can be entered in either rectangular or polar form.

```
a + i b a + ib \exp(1/3 \text{ i pi}) \exp\left(\frac{1}{3}i\pi\right)
```

Converting a complex number to rectangular or polar coordinates causes simplification of mixed forms.

```
A = 1 + i

B = sqrt(2) exp(1/4 i pi)

A - B

1 + i - 2^{1/2} \exp\left(\frac{1}{4}i\pi\right)
rect(last)
```

Rectangular complex quantities, when raised to a power, are multiplied out.

$$(a + i b)^2$$

$$a^2 - b^2 + 2iab$$

When a and b are numerical and the power is negative, the evaluation is done as follows.

$$(a+ib)^{-n} = \left(\frac{a-ib}{(a+ib)(a-ib)}\right)^n = \left(\frac{a-ib}{a^2+b^2}\right)^n$$

Here are a few examples.

$$1/(2 - i)$$

$$\frac{2}{5} + \frac{1}{5}i$$
 $(-1 + 3 i)/(2 - i)$ 

The absolute value of a complex number returns its magnitude.

```
abs(3 + 4 i)
```

-1+i

5

The imaginary unit can be changed from i to j by defining  $j = \sqrt{-1}$ .

```
j = sqrt(-1)
sqrt(-4)
```

2j

### 1.8 Linear algebra

The dot function is used to multiply vectors, matrices, and tensors. For example, let

$$A = \begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}, \quad x = \begin{pmatrix} x_1 \\ x_2 \end{pmatrix}$$

The product Ax is computed as follows.

$$A = ((1,2),(3,4))$$

$$x = (x1,x2)$$

$$dot(A,x)$$

$$\begin{bmatrix} x_1 + 2x_2 \\ 3x_1 + 4x_2 \end{bmatrix}$$

The following example shows how to use dot and inv to solve for vector X in AX = B.

A = ((3,7),(1,-9))  
B = (16,-22)  
X = dot(inv(A),B)  
X  

$$X = \begin{bmatrix} -\frac{5}{17} \\ \frac{41}{17} \end{bmatrix}$$

The dot function can have more than two arguments. For example, dot(A,B,C) can be used for the dot product of three tensors.

Square brackets are used for component access. Index numbering starts with 1.

A = ((a,b),(c,d))  
A[1,2] = -A[1,1]  
A  
$$\begin{bmatrix} a & -a \\ c & d \end{bmatrix}$$

The following example demonstrates the relation  $A^{-1} = (\det A)^{-1} \operatorname{adj} A$ .

```
A = ((a,b),(c,d))
inv(A) == adj(A) / det(A)
1
```

Sometimes a calculation will be simpler if it can be reorganized to use adj instead of inv. The main idea is to try to prevent the determinant from appearing as a divisor. For example, suppose for matrices A and B you want to show that

$$A - B^{-1} = 0$$

Depending on the complexity of  $\det B$ , the software may not be able to find a simplification that yields zero. Should that occur, the following alternative formulation can be tried.

$$A \det B - \operatorname{adj} B = 0$$

### 1.9 Component arithmetic

Tensor plus scalar adds the scalar to each component of the tensor.

$$\begin{bmatrix} x + 10 \\ y + 10 \\ z + 10 \end{bmatrix}$$

The product of two tensors is the Hadamard (element-wise) product.

```
A = ((1,2),(3,4))

B = ((a,b),(c,d))

A B
\begin{bmatrix} a & 2b \\ 3c & 4d \end{bmatrix}
```

Tensor raised to a power raises each component to the power.

$$\begin{pmatrix} \mathbf{x}, \mathbf{y}, \mathbf{z} \end{pmatrix} \hat{}_{2}^{2}$$
$$\begin{pmatrix} x^{2} \\ y^{2} \\ z^{2} \end{pmatrix}$$

# 2 Calculus

#### 2.1 Derivative

 $\mathtt{d}(f,x)$  returns the derivative of f with respect to x. The x can be omitted for expressions in x.

$$d(x^2)$$

2x

The following table summarizes the various ways to obtain multi-derivatives.

$$\frac{\partial^2 f}{\partial x^2} \qquad \text{d(f,x,x)} \qquad \text{d(f,x,2)}$$

$$\frac{\partial^2 f}{\partial x \partial y} \qquad \text{d(f,x,y)}$$

$$\frac{\partial^{m+n+\cdots} f}{\partial x^m \partial y^n \cdots} \qquad \text{d(f,x,m,y,n,...)}$$

#### 2.2 Gradient

The gradient of f is obtained by using a vector for x in d(f, x).

r = sqrt(x^2 + y^2)  
d(r,(x,y))  
$$\begin{bmatrix} \frac{x}{(x^2 + y^2)^{1/2}} \\ \frac{y}{(x^2 + y^2)^{1/2}} \end{bmatrix}$$

The f in d(f, x) can be a vector or higher order function. Gradient raises the rank by one.

$$F = (x + 2 y, 3 x + 4 y)$$

$$X = (x,y)$$

$$d(F,X)$$

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

### 2.3 Template functions

Function f in d(f, x) does not have to be defined. Function f can be a template function with just a name and an argument list. The argument list determines the result of d. For example, d(f(x), x) evaluates to itself because f depends on x. However, d(f(x), y) evaluates to zero because f does not depend on g.

```
d(f(x),x) --- f(x) depends on x d(f(x),x) d(f(x),y) --- f(x) does not depend on y 0 d(f(x,y),y) --- f(x,y) depends on both x and y d(f(x,y),y) d(f(x,y),y) d(f(x,y),y)
```

As the final example shows, an empty argument list causes d to evaluate to itself, regardless of the second argument.

Template functions are useful for experimenting with differential forms. For example, let us check the identity

$$\operatorname{div}(\operatorname{curl} F) = 0$$

for an arbitrary vector function F.

0

```
F = (F1(x,y,z),F2(x,y,z),F3(x,y,z))
div(curl(F))
```

### 2.4 Integral

integral(f, x) returns the integral of f with respect to x. The x can be omitted for expressions in x. The argument list can be extended for multiple integrals.

```
integral(x^2) \frac{1}{3}x^3 integral(x y,x,y) \frac{1}{4}x^2y^2
```

 $\operatorname{defint}(f, x, a, b, \ldots)$  computes the definite integral of f with respect to x evaluated from a to b. The argument list can be extended for multiple integrals. The following example computes the integral of  $f = x^2$  over the domain of a semicircle. For each x along the abscissa, y ranges from 0 to  $\sqrt{1-x^2}$ .

```
defint(x^2,y,0,sqrt(1 - x^2),x,-1,1) \frac{1}{8}\pi
```

Alternatively, the eval function can be used to compute a definite integral step by step.

```
I = integral(x^2, y)
I = eval(I, y, sqrt(1 - x^2)) - eval(I, y, 0)
I = integral(I, x)
eval(I, x, 1) - eval(I, x, -1)
\frac{1}{8}\pi
```

Here is a useful trick. Integrals involving sine and cosine can often be solved by using exponentials. Trigonometric simplifications involving powers and multiple angles turn into simple algebra in the exponential domain. For example, the definite integral

$$\int_0^{2\pi} (\sin^4 t - 2\cos^3(t/2)\sin t) dt$$

can be solved as follows.

```
f = \sin(t)^4 - 2 \cos(t/2)^3 \sin(t)
f = \operatorname{circexp}(f)
defint(f,t,0,2 pi)
\frac{3}{4}\pi - \frac{16}{5}
```

### 2.5 Arc length

Let g(t) be a function that draws a curve. The arc length from g(a) to g(b) is given by

$$\int_{a}^{b} |g'(t)| dt$$

where |g'(t)| is the length of the tangent vector at g(t). The integral sums over all of the tangent lengths to arrive at the total length from a to b. For example, let us measure the length of  $x^2$  from x = 0 to x = 1. A suitable g(t) for the arc  $x^2$  is

$$g(t) = (t, t^2), \quad 0 \le t \le 1$$

Hence one Eigenmath solution for computing the arc length is

```
 \begin{aligned} & x = t \\ & y = t^2 \\ & g = (x,y) \\ & \text{defint(abs(d(g,t)),t,0,1)} \\ & \frac{1}{2} \, 5^{1/2} + \frac{1}{4} \log(5^{1/2} + 2) \\ & \text{float} \end{aligned}
```

1.47894

As expected, the result is greater than  $\sqrt{2} \approx 1.414$ , the length of the diagonal from (0,0) to (1,1).

The result seems rather complicated given that we started with a simple parabola. Let us inspect |g'(t)| to see why.

g

```
g = \begin{bmatrix} t \\ t^2 \end{bmatrix}
d(g,t)
\begin{bmatrix} 1 \\ 2t \end{bmatrix}
abs(d(g,t))
(4t^2 + 1)^{1/2}
```

The following script does a discrete computation of the arc length by dividing the curve into 100 pieces.

```
g(t) = (t,t^2)

h(k) = abs(g(k/100.0) - g((k-1)/100.0))

sum(k,1,100,h(k))
```

1.47894

As expected, the discrete result matches the analytic result.

Find the length of the curve  $y = x^{3/2}$  from the origin to  $x = \frac{4}{3}$ .

```
x = t

y = x^{(3/2)}

g = (x,y)

defint(abs(d(g,x)),x,0,4/3)
```

Because of the way t is substituted for x, the following code yields the same result.

```
g = (t,t^{(3/2)})

defint(abs(d(g,t)),t,0,4/3)

\frac{56}{27}
```

### 2.6 Line integrals

There are two different kinds of line integrals, one for scalar fields and one for vector fields. The following table shows how both are based on the calculation of arc length.

	Abstract form	Computable form
Arc length	$\int_C ds$	$\int_{a}^{b}  g'(t)  dt$
Line integral, scalar field	$\int_C f  ds$	$\int_{a}^{b} f(g(t))  g'(t)  dt$
Line integral, vector field	$\int_C (F \cdot u)  ds$	$\int_{a}^{b} F(g(t)) \cdot g'(t)  dt$

For the vector field form, the symbol u is the unit tangent vector

$$u = \frac{g'(t)}{|g'(t)|}$$

The length of the tangent vector cancels with ds as follows.

$$\int_C (F \cdot u) \, ds = \int_a^b \left( F(g(t)) \cdot \frac{g'(t)}{|g'(t)|} \right) \left( |g'(t)| \, dt \right) = \int_a^b F(g(t)) \cdot g'(t) \, dt$$

Evaluate

$$\int_C x \, ds$$
 and  $\int_C x \, dx$ 

where C is a straight line from (0,0) to (1,1).

What a difference the measure makes. The first integral is over a scalar field and the second is over a vector field. This can be understood when we recall that

$$ds = |g'(t)| dt$$

Hence for  $\int_C x \, ds$  we have

```
x = t
y = t
g = (x,y)
defint(x abs(d(g,t)),t,0,1)
\frac{}{2^{1/2}}
For \int_C x \, dx we have
x = t
y = t
g = (x,y)
F = (x,0)
defint(dot(F,d(g,t)),t,0,1)
\frac{1}{2}
The following line integral problems are from Advanced Calculus, Fifth Edi-
tion by Wilfred Kaplan.
Evaluate \int y^2 dx along the straight line from (0,0) to (2,2).
x = 2t
y = 2t
g = (x,y)
F = (y^2, 0)
defint(dot(F,d(g,t)),t,0,1)
Evaluate \int z \, dx + x \, dy + y \, dz along the path x = 2t + 1, y = t^2, z = 1 + t^3,
0 \le t \le 1.
x = 2t+1
y = t^2
z = 1+t^3
g = (x,y,z)
F = (z, x, y)
defint(dot(F,d(g,t)),t,0,1)
```

#### 2.7 Surface area

Let S be a surface parameterized by x and y. That is, let S = (x, y, z) where z = f(x, y). The tangent lines at a point on S form a tiny parallelogram. The area a of the parallelogram is given by the magnitude of the cross product.

$$a = \left| \frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y} \right|$$

By summing over all the parallelograms we obtain the total surface area A. Hence

$$A = \int \int dA = \int \int a \, dx \, dy$$

The following example computes the surface area of a unit disk parallel to the xy plane.

```
z = 2

S = (x,y,z)

a = abs(cross(d(S,x),d(S,y)))

defint(a,y,-sqrt(1 - x^2),sqrt(1 - x^2),x,-1,1)
```

The result is  $\pi$ , the area of a unit circle, which is what we expect. The following example computes the surface area of  $z = x^2 + 2y$  over a unit square.

```
z = x^2 + 2y

S = (x,y,z)

a = abs(cross(d(S,x),d(S,y)))

defint(a,x,0,1,y,0,1)

\frac{5}{8}\log(5) + \frac{3}{2}
```

The following exercise is from *Multivariable Mathematics* by Williamson and Trotter, p. 598. Find the area of the spiral ramp defined by

$$S = \begin{bmatrix} u \cos v \\ u \sin v \\ v \end{bmatrix}, \quad 0 \le u \le 1, \quad 0 \le v \le 3\pi$$

```
 \begin{array}{l} x = u \, \cos(v) \\ y = u \, \sin(v) \\ z = v \\ S = (x,y,z) \\ a = circexp(abs(cross(d(S,u),d(S,v)))) \\ defint(a,u,0,1,v,0,3pi) \\ \\ \frac{3\pi}{2^{1/2}} + \frac{3}{2}\pi \log \left(2^{1/2} + 1\right) \\ \\ \text{float} \end{array}
```

10.8177

### 2.8 Surface integrals

A surface integral is like adding up all the wind on a sail. In other words, we want to compute

$$\iint \mathbf{F} \cdot \mathbf{n} \, dA$$

where  $\mathbf{F} \cdot \mathbf{n}$  is the amount of wind normal to a tiny parallelogram dA. The integral sums over the entire area of the sail. Let S be the surface of the sail parameterized by x and y. (In this model, the z direction points downwind.) By the properties of the cross product we have the following for the unit normal  $\mathbf{n}$  and for dA.

$$\mathbf{n} = \frac{\frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y}}{\left| \frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y} \right|} \qquad dA = \left| \frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y} \right| dx dy$$

Hence

$$\iint \mathbf{F} \cdot \mathbf{n} \, dA = \iint \mathbf{F} \cdot \left( \frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y} \right) \, dx \, dy$$

The following exercise is from *Advanced Calculus* by Wilfred Kaplan, p. 313. Evaluate the surface integral

$$\iint_{S} \mathbf{F} \cdot \mathbf{n} \, d\sigma$$

where  $\mathbf{F} = xy^2z\mathbf{i} - 2x^3\mathbf{j} + yz^2\mathbf{k}$ , S is the surface  $z = 1 - x^2 - y^2$ ,  $x^2 + y^2 \le 1$  and **n** is upper.

Note that the surface intersects the xy plane in a circle. By the right hand rule, crossing x into y yields  $\mathbf{n}$  pointing upwards hence

$$\mathbf{n} \, d\sigma = \left(\frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y}\right) \, dx \, dy$$

The following code computes the surface integral. The symbols f and h are used as temporary variables.

```
z = 1 - x^2 - y^2

F = (x y^2 z, -2 x^3, y z^2)

S = (x, y, z)

f = dot(F, cross(d(S, x), d(S, y)))

h = sqrt(1 - x^2)

defint(f, y, -h, h, x, -1, 1)
```

#### 2.9 Green's theorem

Green's theorem tells us that

$$\oint P dx + Q dy = \iint \left( \frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right) dx dy$$

In other words, a line integral and a surface integral can yield the same result.

Example 1. The following exercise is from Advanced Calculus by Wilfred Kaplan, p. 287. Evaluate  $\oint (2x^3 - y^3) dx + (x^3 + y^3) dy$  around the circle  $x^2 + y^2 = 1$  using Green's theorem.

Use polar coordinates to solve.

```
P = 2x^3 - y^3
Q = x^3 + y^3
f = d(Q,x) - d(P,y)
x = r cos(theta)
y = r sin(theta)
defint(f r,r,0,1,theta,0,2pi)
```

```
\frac{3}{2}\pi
```

The defint integrand is f r because  $r dr d\theta = dx dy$ .

Now let us try computing the line integral side of Green's theorem and see if we get the same result. We need to use the trick of converting sine and cosine to exponentials so that Eigenmath can find a solution.

```
x = cos(t)

y = sin(t)

P = 2x^3 - y^3

Q = x^3 + y^3

f = P d(x,t) + Q d(y,t)

f = circexp(f)

defint(f,t,0,2pi)
```

Example 2. Compute both sides of Green's theorem for F=(1-y,x) over the disk  $x^2+y^2\leq 4$ .

First compute the line integral along the boundary of the disk. Note that the radius of the disk is 2.

```
-- Line integral
P = 1 - y
Q = x
x = 2 cos(t)
y = 2 sin(t)
defint(P d(x,t) + Q d(y,t),t,0,2pi)
8π

-- Surface integral
x = quote(x) -- clear x
y = quote(y) -- clear y
h = sqrt(4-x^2)
defint(d(Q,x) - d(P,y),y,-h,h,x,-2,2)
```

```
-- use polar coordinates 

P = 1 - y 

Q = x 

f = d(Q,x) - d(P,y) -- do before change of coordinates 

x = r \cos(\text{theta}) 

y = r \sin(\text{theta}) 

defint(f r,r,0,2,\text{theta},0,2pi) 

8\pi 

defint(f r,\text{theta},0,2pi,r,0,2) -- integrate over theta first 

8\pi
```

In this case, Eigenmath solved both forms of the polar integral. However, in cases where Eigenmath fails to solve a double integral, try changing the order of integration.

#### 2.10 Stokes's theorem

Stokes's theorem equates a surface integral of the curl of a function with a line integral of the same function. In rectangular coordinates the equivalence is

$$\iint_{S} (\operatorname{curl} \mathbf{F}) \cdot \mathbf{n} \, d\sigma = \oint (P \, dx + Q \, dy + R \, dz)$$

where  $\mathbf{F} = (P, Q, R)$ . For S parametrized by x and y we have

$$\mathbf{n} d\sigma = \left(\frac{\partial S}{\partial x} \times \frac{\partial S}{\partial y}\right) dx dy$$

For example, let  $\mathbf{F} = (y, z, x)$  and let S be the part of the paraboloid  $z = 4 - x^2 - y^2$  that is above the xy plane. The perimeter of the paraboloid is the circle  $x^2 + y^2 = 2$ . The following script computes both the surface and line integrals. Polar coordinates are used for the line integral so that **defint** can succeed.

```
"Surface integral"

z = 4 - x^2 - y^2

F = (y,z,x)
```

```
S = (x, y, z)
z = quote(z) -- clear z for use by curl
f = dot(curl(F),cross(d(S,x),d(S,y)))
x = r \cos(theta)
y = r \sin(theta)
defint(f r,r,0,2,theta,0,2pi)
"Line integral"
x = 2 \cos(t)
y = 2 \sin(t)
z = 4 - x^2 - y^2
P = y
Q = z
R = x
f = P d(x,t) + Q d(y,t) + R d(z,t)
f = circexp(f)
defint(f,t,0,2pi)
```

This is the result when the script runs. Both the surface integral and the line integral yield the same result.

```
Surface integral -4\pi Line integral -4\pi
```

# 3 Quantum Computing

A quantum computer can be simulated by applying rotations to a unit vector  $u \in \mathbb{C}^{2^n}$  where  $\mathbb{C}$  is the set of complex numbers and n is the number of qubits. The dimension is  $2^n$  because a register with n qubits has  $2^n$  eigenstates. (Recall that an eigenstate is the output of a quantum computer.) Quantum operations are "rotations" because they preserve |u| = 1. Mathematically, a rotation of u is equivalent to the product Ru where R is a  $2^n \times 2^n$  matrix.

Eigenstates  $|j\rangle$  are represented by the following vectors. (Each vector has  $2^n$  elements.)

$$|0\rangle = (1, 0, 0, \dots, 0)$$

$$|1\rangle = (0, 1, 0, \dots, 0)$$

$$|2\rangle = (0, 0, 1, \dots, 0)$$

$$\vdots$$

$$|2^{n} - 1\rangle = (0, 0, 0, \dots, 1)$$

A quantum computer algorithm is a sequence of rotations applied to the initial state  $|0\rangle$ . (The sequence could be combined into a single rotation by associativity of matrix multiplication.) Let  $\psi_f$  be the final state of the quantum computer after all the rotations have been applied. Like any other state,  $\psi_f$  is a linear combination of eigenstates.

$$\psi_f = \sum_{j=0}^{2^n - 1} c_j |j\rangle, \quad c_j \in \mathbb{C}, \quad |\psi_f| = 1$$

The last step is to measure  $\psi_f$  and get a result. Measurement rotates  $\psi_f$  to an eigenstate  $|j\rangle$ . The measurement result is  $|j\rangle$ . The probability  $P_j$  of getting a specific result  $|j\rangle$  is

$$P_j = |c_j|^2 = c_j c_j^*$$

Note that if  $\psi_f$  is already an eigenstate then no rotation occurs. (The probability of observing a different eigenstate is zero.) Since the measurement result is always an eigenstate, the coefficients  $c_j$  cannot be observed. However, the same calculation can be run multiple times to obtain a probability

distribution of results. The probability distribution is an estimate of  $|c_j|^2$  for each  $|j\rangle$  in  $\psi_f$ .

Unlike a real quantum computer, in a simulation the final state  $\psi_f$ , or any other state, is available for inspection. Hence there is no need to simulate the measurement process. The probability distribution of the result can be computed directly as

$$P = \psi_f \, \psi_f^*$$

where  $\psi_f \psi_f^*$  is the Hadamard (element-wise) product of vector  $\psi_f$  and its complex conjugate. Result P is a vector such that  $P_j$  is the probability of eigenstate  $|j\rangle$  and

$$\sum_{j=0}^{2^n-1} P_j = 1$$

Note: Eigenmath index numbering begins with 1 hence P[1] is the probability of  $|0\rangle$ , P[2] is the probability of  $|1\rangle$ , etc.

The Eigenmath function rotate(u, s, k, ...) rotates vector u and returns the result. Vector u is required to have  $2^n$  elements where n is an integer from 1 to 15. Arguments s, k, ... are a sequence of rotation codes where s is an upper case letter and k is a qubit number from 0 to n-1. Rotations are evaluated from left to right. The available rotation codes are

C, k Control prefix

H, k Hadamard

 $P, k, \phi$  Phase modifier (use  $\phi = \frac{1}{4}\pi$  for T rotation)

Q, k Quantum Fourier transform

V, k Inverse quantum Fourier transform

W, k, j Swap bits

X, k Pauli X

Y, k Pauli Y

Z, k Pauli Z

Control prefix C, k modifies the next rotation code so that it is a controlled rotation with k as the control qubit. Use two or more prefixes to specify multiple control qubits. For example, C, k, C, j, X, m is a Toffoli rotation. Fourier rotations Q, k and V, k are applied to qubits 0 through k. (Q and V ignore any control prefix.)

List of rotate(u, s, k, ...) error codes:

- 1 Argument u is not a vector or does not have  $2^n$  elements where  $n = 1, 2, \ldots, 15$ .
- 2 Unexpected end of argument list (i.e., missing argument).
- 3 Bit number format error or range error.
- 4 Unknown rotation code.

Example: Verify the following truth table for quantum operator CNOT where qubit 0 is the control and qubit 1 is the target. (Target is inverted when control is set.)

Target	Control	Output
0	0	00
0	1	11
1	0	10
1	1	01

```
U(psi) = rotate(psi,C,0,X,1) -- CNOT, control 0, target 1
ket00 = (1,0,0,0)
ket01 = (0,1,0,0)
ket10 = (0,0,1,0)
ket11 = (0,0,0,1)

U(ket00) == ket00
U(ket01) == ket11
U(ket10) == ket10
U(ket11) == ket01
```

Here are some useful Eigenmath code snippets for setting up a simulation and computing the result.

1. Initialize  $\psi = |0\rangle$ .

```
n = 4 -- number of qubits (example)

N = 2^n -- number of eigenstates

psi = zero(N)

psi[1] = 1
```

2. Compute the probability distribution for state  $\psi$ .


```
P = psi conj(psi)
Hence
               P[1] = \text{probability that } |0\rangle \text{ will be the result}
               P[2] = \text{probability that } |1\rangle \text{ will be the result}
               P[3] = \text{probability that } |2\rangle \text{ will be the result}
               P[N] = \text{probability that } |N-1\rangle \text{ will be the result}
3. (macOS) Draw a probability distribution.
xrange = (0,N)
yrange = (0,1)
draw(P[ceiling(x)],x)
4. Compute an expectation value.
sum(k,1,N, (k-1) P[k])
5. Make the high order qubit "don't care."
for(k,1,N/2, P[k] = P[k] + P[k + N/2])
Hence for N = 16
              P[1] = \text{probability that the result will be } |0\rangle \text{ or } |8\rangle
              P[2] = \text{probability that the result will be } |1\rangle \text{ or } |9\rangle
              P[3] = \text{probability that the result will be } |2\rangle \text{ or } |10\rangle
```

 $P[8] = \text{probability that the result will be } |7\rangle \text{ or } |15\rangle$ 

# 4 Draw (macOS)


 $\mathtt{draw}(f,x)$  draws a graph of function f of x. The default second argument is symbol  $\mathtt{x}$ .

```
draw(x^2)
```




The vectors xrange and yrange control the scale of the graph.

```
xrange = (-1,1)
yrange = (0,2)
draw(x^2)
```



Parametric drawing occurs when a function returns a vector. The vector trange controls the parametric range. The default is trange=(-pi,pi). In the following example, draw varies theta over the default range  $-\pi$  to  $+\pi$ .

```
xrange = (-10,10)
yrange = (-10,10)
f = 5 (cos(theta),sin(theta))
draw(f,theta)
```




In the following example, trange is reduced to draw a quarter circle instead of a full circle.

```
trange = (0,pi/2)
f = 5 (cos(theta),sin(theta))
draw(f,theta)
```




Draw a lemniscate.

```
trange = (-pi,pi)
X = cos(t) / (1 + sin(t)^2)
Y = sin(t) cos(t) / (1 + sin(t)^2)
f = 5 (X,Y)
draw(f,t)
```



Draw a cardioid.

```
r = (1 + cos(t)) / 2
u = (cos(t),sin(t))
f = r u
xrange = (-1,1)
yrange = (-1,1)
trange = (0,2pi)
draw(f,t)
```



# 5 Function Reference

# abs(x)

Returns the absolute value or vector length of x.

```
X = (x,y,z)
abs(X)
(x^2 + y^2 + z^2)^{1/2}
```

# adj(m)

Returns the adjunct of matrix m. Adjunct is equal to determinant times inverse.

```
A = ((a,b),(c,d))
adj(A) == det(A) inv(A)
```

 $and(a, b, \ldots)$ 

Returns 1 if all arguments are true (nonzero). Returns 0 otherwise.

```
and(1=1,2=2)
```

1

1

# arccos(x)

Returns the arc cosine of x.

```
arccos(1/2)
```

 $\frac{1}{3}\pi$ 

# $\operatorname{arccosh}(x)$

Returns the arc hyperbolic cosine of x.

# $\arcsin(x)$

Returns the arc sine of x.

```
arcsin(1/2)
```

 $\frac{1}{6}\pi$ 

## $\operatorname{arcsinh}(x)$

Returns the arc hyperbolic sine of x.

# $\arctan(y, x)$

Returns the arc tangent of y over x. If x is omitted then x = 1 is used.

```
arctan(1,0)
```

 $\frac{1}{2}\pi$ 

# $\operatorname{arctanh}(x)$

Returns the arc hyperbolic tangent of x.

# arg(z)

Returns the angle of complex z.

```
arg(2 - 3i)
```

 $\arctan(-3,2)$ 

# binding(s)

The result of evaluating a symbol can differ from the symbol's binding. For example, the result may be expanded. The binding function returns the actual binding of a symbol.

```
p = quote((x + 1)^2)
```

```
p = x^2 + 2x + 1
binding(p)
(x+1)^2
```

# $\mathbf{ceiling}(x)$

Returns the smallest integer greater than or equal to x.

```
ceiling(1/2)
```

1

# $\mathbf{check}(x)$

If x is true (nonzero) then continue, else stop. Expression x can include the relational operators =, ==, <, <=, >, >=. Use the not function to test for inequality.

```
A = \exp(i pi)

B = -1

\operatorname{check}(A == B) -- stop here if A not equal to B
```

# circexp(x)

Returns expression x with circular and hyperbolic functions converted to exponentials.

```
circexp(cos(x) + i sin(x))
exp(ix)
```

#### clear

Clears all symbol definitions.

# $\operatorname{clock}(z)$

Returns complex z in polar form with base of negative 1 instead of e.

```
clock(2 - 3i) 13^{1/2} (-1)^{\arctan(-3,2)/\pi}
```

# $\mathbf{cofactor}(m, i, j)$

Returns the cofactor of matrix m for row i and column j.

```
A = ((a,b),(c,d))
cofactor(A,1,2) == adj(A)[2,1]
1
```

# conj(z)

Returns the complex conjugate of z.

```
conj(2 - 3i)
 2 + 3i
```

# contract(a, i, j)

Returns tensor a summed over indices i and j. If i and j are omitted then 1 and 2 are used. The expression contract(m) computes the trace of matrix m.

```
A = ((a,b),(c,d))contract(A)
```

# $\cos(x)$

a + d

Returns the cosine of x.

```
cos(pi/4)

\frac{1}{2^{1/2}}
```

#### $\cosh(x)$

Returns the hyperbolic cosine of x.

```
circexp(cosh(x))
\frac{1}{2}\exp(-x) + \frac{1}{2}\exp(x)
```

## $\mathbf{d}(f,x)$

Returns the partial derivative of f with respect to x.

```
d(x^2,x)
```

2x

Argument f can be a tensor of any rank. Argument x can be a vector. When x is a vector the result is the gradient of f.

```
F = (f(),g(),h())
X = (x,y,z)
d(F,X)
\begin{bmatrix} d(f(),x) & d(f(),y) & d(f(),z) \\ d(g(),x) & d(g(),y) & d(g(),z) \\ d(h(),x) & d(h(),y) & d(h(),z) \end{bmatrix}
```

It is ok to use d as a variable name. It will not conflict with function d.

It is ok to redefine d as a different function. The function derivative, a synonym for d, can still be used to obtain a partial derivative.

## $\mathbf{defint}(f, x, a, b)$

Returns the definite integral of f with respect to x evaluated from a to b. The argument list can be extended for multiple integrals as shown in the following example.

```
f = (1 + cos(theta)^2) sin(theta) -- integrate over theta then over phi defint(f, theta, 0, pi, phi, 0, 2 pi) \frac{16}{3}\pi
```

### denominator(x)

Returns the denominator of expression x.

```
denominator(a/b)
b
```

# det(m)

Returns the determinant of matrix m.

```
A = ((a,b),(c,d))
det(A)
ad - bc
```

# dim(a, n)

Returns the dimension of the nth index of tensor a. Index numbering starts with 1.

```
A = ((1,2),(3,4),(5,6))

dim(A,1)
```

3

$$do(a, b, \ldots)$$

Evaluates each argument from left to right. Returns the result of the final argument.

```
do(A=1,B=2,A+B)
```

### $dot(a, b, \ldots)$

Returns the dot product of vectors, matrices, and tensors. Also known as the matrix product. Arguments are evaluated from right to left. The following example solves for X in AX = B.

```
A = ((1,2),(3,4))

B = (5,6)

X = dot(inv(A),B)

X
\begin{bmatrix} -4 \\ \frac{9}{2} \end{bmatrix}
```

#### eigenvec(m)

Returns eigenvectors for matrix m. Matrix m is required to be numerical, real, and symmetric. The return value is a matrix with each column an eigenvector. Eigenvalues are obtained as shown.

```
A = ((1,2,3),(2,6,4),(3,4,5))
Q = eigenvec(A)
D = dot(transpose(Q),A,Q) -- eigenvalues on the diagonal of D dot(Q,D,transpose(Q))
\begin{bmatrix} 1 & 2 & 3 \\ 2 & 6 & 4 \\ 3 & 4 & 5 \end{bmatrix}
```

eval(f, x, a)

Returns expression f evaluated at x equals a. The argument list can be extended as shown.

```
f = sqrt(x^2 + y^2)
eval(f,x,3,y,4) -- evaluate f at x=3 and y=4
```

In the following example, eval is used to compute

$$\frac{d^{l+m}}{d(\cos\theta)^{l+m}}(\cos^2\theta - 1)^l$$

-- associated legendre of cos theta

$$P(1,m) = test(m < 0, (-1)^m (1 + m)! / (1 - m)! P(1,-m), 1 / (2^1 1!)  $sin(theta)^m * eval(d((x^2 - 1)^1,x,1 + m),x,cos(theta)))$$$

P(2,-1)

$$-\frac{1}{2}\cos(\theta)\sin(\theta)$$

#### $\exp(x)$

Returns the exponential of x.

exp(i pi)

-1

## $\exp\cos(z)$

Returns the cosine of z in exponential form.

#### expcos(z)

$$\frac{1}{2}\exp(iz) + \frac{1}{2}\exp(-iz)$$

#### expcosh(z)

Returns the hyperbolic cosine of z in exponential form.

#### expcosh(z)

$$\frac{1}{2}\exp(-z) + \frac{1}{2}\exp(z)$$

### expsin(z)

Returns the sine of z in exponential form.

#### expsin(z)

$$-\frac{1}{2}i\exp(iz) + \frac{1}{2}i\exp(-iz)$$

#### expsinh(z)

Returns the hyperbolic sine of z in exponential form.

#### expsinh(z)

$$-\frac{1}{2}\exp(-z) + \frac{1}{2}\exp(z)$$

#### exptan(z)

Returns the tangent of z in exponential form.

#### exptan(z)

$$\frac{i}{\exp(2iz)+1} - \frac{i\exp(2iz)}{\exp(2iz)+1}$$

## exptanh(z)

Returns the hyperbolic tangent of z in exponential form.

#### exptanh(z)

$$-\frac{1}{\exp(2z) + 1} + \frac{\exp(2z)}{\exp(2z) + 1}$$

# factorial(n)

Returns the factorial of n. The expression n! can also be used.

20!

### float(x)

Returns expression x with rational numbers and integers converted to floating point values. The symbol pi and the natural number are also converted.

```
float(212^17)
3.52947 \times 10^{39}
```

## floor(x)

Returns the largest integer less than or equal to x.

```
floor(1/2)
```

0

$$\mathbf{for}(i, j, k, a, b, \ldots)$$

For i equals j through k evaluate a, b, etc.

```
for(k,1,3,A=k,print(A))
```

A = 1

A = 2

A = 3

Note: The original value of i is restored after for completes. If symbol i is used for index variable i then the imaginary unit is overridden in the scope of for.

## hadamard(a, b, ...)

Returns the Hadamard (element-wise) product.

$$\begin{bmatrix} a^2 \\ b^2 \\ c^2 \end{bmatrix}$$

#### i

Symbol i is initialized to the imaginary unit  $\sqrt{-1}$ .

```
exp(i pi)
```

-1

Note: It is ok to clear or redefine i and use the symbol for something else.

#### imag(z)

Returns the imaginary part of complex z.

```
imag(2 - 3i)
-3
```

#### infixform(x)

Converts expression x to a string and returns the result.

```
p = (x + 1)^2

infixform(p)

x^2 + 2 x + 1
```

 $inner(a, b, \ldots)$ 

Returns the inner product of vectors, matrices, and tensors. Also known as the matrix product.

```
A = ((a,b),(c,d))
B = (x,y)
inner(A,B)
\begin{bmatrix} ax + by \\ cx + dy \end{bmatrix}
```

Note: inner and dot are the same function.

## integral(f, x)

Returns the integral of f with respect to x.

```
integral(x^2,x)
```

 $\frac{1}{3}x^{3}$ 

## inv(m)

Returns the inverse of matrix m.

$$A = ((1,2),(3,4))$$

inv(A)

$$\begin{bmatrix} -2 & 1 \\ \frac{3}{2} & -\frac{1}{2} \end{bmatrix}$$

j

Set j=sqrt(-1) to use j for the imaginary unit instead of i.

```
j = sqrt(-1)
1/sqrt(-1)
```

-j

# kronecker(a, b, ...)

Returns the Kronecker product of vectors and matrices.

$$A = ((1,2),(3,4))$$

B = ((a,b),(c,d))

kronecker(A,B)

$$\begin{bmatrix} a & b & 2a & 2b \\ c & d & 2c & 2d \\ 3a & 3b & 4a & 4b \\ 3c & 3d & 3c & 4d \end{bmatrix}$$

#### last

The result of the previous calculation is stored in last.

#### 212^17

3529471145760275132301897342055866171392

last^(1/17)

212

Symbol last is an implied argument when a function has no argument list.

#### 212^17

3529471145760275132301897342055866171392

#### float

 $3.52947 \times 10^{39}$ 

#### $\log(x)$

Returns the natural logarithm of x.

#### log(x^y)

 $y \log(x)$ 

# mag(z)

Returns the magnitude of complex z. Function mag treats undefined symbols as real while abs does not.

$$mag(x + i y)$$

$$(x^2+y^2)^{1/2}$$

```
minor(m, i, j)
```

Returns the minor of matrix m for row i and column j.

```
A = ((1,2,3),(4,5,6),(7,8,9))
minor(A,1,1) == det(minormatrix(A,1,1))
1
```

### minormatrix(m, i, j)

Returns a copy of matrix m with row i and column j removed.

```
A = ((1,2,3),(4,5,6),(7,8,9))
minormatrix(A,1,1)
\begin{bmatrix} 5 & 6 \\ 8 & 9 \end{bmatrix}
```

### noexpand(x)

Evaluates expression x without expanding products of sums.

```
noexpand((x + 1)^2 / (x + 1)) x + 1
```

# $\mathbf{not}(x)$

Returns 0 if x is true (nonzero). Returns 1 otherwise.

```
not(1=1)
```

## $\mathbf{nroots}(p, x)$

Returns the approximate roots of polynomials with real or complex coefficients. Multiple roots are returned as a vector.

```
p = x^5 - 1
nroots(p,x)
\begin{bmatrix}
1 \\
-0.809017 + 0.587785i \\
-0.809017 - 0.587785i \\
0.309017 + 0.951057i \\
0.309017 - 0.951057i
\end{bmatrix}
```

#### numerator(x)

Returns the numerator of expression x.

```
numerator(a/b)
a
```

$$or(a, b, \ldots)$$

Returns 1 if at least one argument is true (nonzero). Returns 0 otherwise.

```
or(1=1,2=2)
1
```

# $\mathbf{outer}(a,b,\ldots)$

Returns the outer product of vectors, matrices, and tensors.

```
A = (a,b,c)
B = (x,y,z)
outer(A,B)
\begin{bmatrix} ax & ay & az \\ bx & by & bz \\ cx & cy & cz \end{bmatrix}
```

#### pi

Symbol for  $\pi$ .

```
exp(i pi)
```

-1

# polar(z)

Returns complex z in polar form.

```
polar(x - i y)
```

$$(x^2 + y^2)^{1/2} \exp(i \arctan(-y, x))$$

#### power

Use ^ to raise something to a power. Use parentheses for negative powers.

```
x^{(-2)}
```

 $\frac{1}{x^2}$ 

# $\mathbf{print}(a, b, \ldots)$

Evaluate expressions and print the results. Useful for printing from inside a for loop.

```
for(j,1,3,print(j))
```

- j = 1
- j = 2
- j = 3

### product(i, j, k, f)

For i equals j through k evaluate f. Returns the product of all f.

```
product(j,1,3,x + j)
x^3 + 6x^2 + 11x + 6
```

The original value of i is restored after product completes. If symbol i is used for index variable i then the imaginary unit is overridden in the scope of product.

#### product(y)

Returns the product of components of y.

```
y = (1,2,3,4)
product(y)
```

24

## quote(x)

Returns expression x without evaluating it first.

```
quote((x + 1)^2)
(x+1)^2
```

## rank(a)

Returns the number of indices that tensor a has.

```
A = ((a,b),(c,d))
rank(A)
```

### rationalize(x)

Returns expression x with everything over a common denominator.

```
rationalize(1/a + 1/b + 1/2)
\frac{2a+ab+2b}{2ab}
```

Note: rationalize returns an unexpanded expression. If the result is assigned to a symbol, evaluating the symbol will expand the result. Use binding to retrieve the unexpanded expression.

```
f = rationalize(1/a + 1/b + 1/2)
binding(f)
\frac{2a + ab + 2b}{2ab}
real(z)
Returns the real part of complex z.
```

```
real(2 - 3i)
```

2

# rect(z)

Returns complex z in rectangular form.

```
rect(exp(i x))
\cos(x) + i\sin(x)
```

## $\mathbf{roots}(p, x)$

Returns the rational roots of a polynomial. Multiple roots are returned as a vector.

```
p = (x + 1) (x - 2)
roots(p,x)
\begin{bmatrix} -1 \\ 2 \end{bmatrix}
```

If no roots are found then nil is returned. A nil result is not printed so the following example uses infixform to print nil as a string.

```
p = x^2 + 1
infixform(roots(p,x))
nil
```

# $rotate(u, s, k, \ldots)$

Rotates vector u and returns the result. Vector u is required to have  $2^n$  elements where n is an integer from 1 to 15. Arguments  $s, k, \ldots$  are a sequence of rotation codes where s is an upper case letter and k is a qubit number from 0 to n-1. Rotations are evaluated from left to right. See section 3 for a list of rotation codes.

```
psi = (1,0,0,0)
rotate(psi,H,0)
```

$$\begin{bmatrix} \frac{1}{2^{1/2}} \\ \frac{1}{2^{1/2}} \\ 0 \\ 0 \end{bmatrix}$$

### run(x)

Run script x where x evaluates to a filename string. Useful for importing function libraries.

```
run("EVA2.txt")
```

For Eigenmath installed from the Mac App Store, run files need to be put in the directory ~/Library/Containers/com.gweigt.eigenmath/Data/

### simplify(x)

Returns expression x in a simpler form.

```
simplify(sin(x)^2 + cos(x)^2)
```

1

## sin(x)

Returns the sine of x.

sin(pi/4)

$$\frac{1}{2^{1/2}}$$

#### sinh(x)

Returns the hyperbolic sine of x.

```
circexp(sinh(x))
```

$$-\frac{1}{2}\exp(-x) + \frac{1}{2}\exp(x)$$

# $\mathbf{sqrt}(x)$

Returns the square root of x.

```
sqrt(10!)
```

 $720 \ 7^{1/2}$ 

#### stop

In a script, it does what it says.

For i equals j through k evaluate f. Returns the sum of all f.

$$sum(j,1,5,x^j)$$

$$x^5 + x^4 + x^3 + x^2 + x$$

The original value of i is restored after sum completes. If symbol i is used for index variable i then the imaginary unit is overridden in the scope of sum.

### sum(y)

Returns the sum of components of y.

$$y = (1,2,3,4)$$
  
sum(y)

10

#### tan(x)

Returns the tangent of x.

$$simplify(tan(x) - sin(x)/cos(x))$$

0

# tanh(x)

Returns the hyperbolic tangent of x.

#### circexp(tanh(x))

$$-\frac{1}{\exp(2x) + 1} + \frac{\exp(2x)}{\exp(2x) + 1}$$

```
\mathbf{test}(a, b, c, d, \ldots)
```

If argument a is true (nonzero) then b is returned, else if c is true then d is returned, etc. If the number of arguments is odd then the final argument is returned if all else fails. Expressions can include the relational operators =, ==, <, <=, >, >=. Use the not function to test for inequality. (The equality operator == is available for contexts in which = is the assignment operator.)

```
A = 1
B = 1
test(A=B, "yes", "no")
yes
```

#### trace

Set trace=1 in a script to print the script as it is evaluated. Useful for debugging.

```
trace = 1
```

Note: The contract function is used to obtain the trace of a matrix.

## transpose(a, i, j)

Returns the transpose of tensor a with respect to indices i and j. If i and j are omitted then 1 and 2 are used. Hence a matrix can be transposed with a single argument.

```
A = ((a,b),(c,d))
transpose(A)
\begin{bmatrix} a & c \\ b & d \end{bmatrix}
```

Note: The argument list can be extended for multiple transpose operations. Arguments are evaluated from left to right. For example, transpose(A,1,2,2,3) is equivalent to

```
transpose(transpose(A,1,2),2,3)
```

# $\mathbf{unit}(n)$

Returns an n by n identity matrix.

#### unit(3)

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\mathbf{zero}(i, j, \ldots)$$

Returns a null tensor with dimensions i, j, etc. Useful for creating a tensor and then setting component values.

$$A = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

#### 6 Tricks

- 1. In a script, line breaking is allowed provided the line breaks occur immediately after operators. The scanner will automatically go to the next line after an operator.
- 2. Setting trace=1 in a script causes each line to be printed just before it is evaluated. This is useful for debugging.
- 3. The last result is stored in the symbol last.
- 4. Use contract(A) to get the mathematical trace of matrix A.
- 5. Use binding(s) to get the unevaluated binding of symbol s.
- 6. Use s=quote(s) to clear symbol s.
- 7. Use float(pi) to get the floating point value of  $\pi$ . Set pi=float(pi) to evaluate expressions with a numerical value for  $\pi$ . Set pi=quote(pi) to make  $\pi$  symbolic again.
- 8. Assign strings to unit names so they are printed normally. For example, setting meter="meter" causes the symbol meter to be printed as meter instead of  $m_{eter}$ .
- 9. Use expsin and expcos instead of sin and cos. Trigonometric simplifications occur automatically when exponentials are used.
- 10. Use A==B or A-B==0 to test for equality of A and B. The equality operator == uses a cross multiply algorithm to eliminate denominators. Hence == can typically determine equality even when the unsimplified result of A-B is nonzero. Note: Equality tests involving floating point numbers can be problematic due to roundoff error.