Green's theorem

This is Green's theorem.

$$\oint (P\,dx + Q\,dy) = \iint \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right) dx\,dy$$

In words, a line integral and a surface integral can yield the same result.

Example 1. The following exercise is from *Advanced Calculus* by Wilfred Kaplan, p. 287. Evaluate $\oint (2x^3 - y^3)\,dx + (x^3 + y^3)\,dy$ around the circle $x^2 + y^2 = 1$ using Green's theorem.

Use polar coordinates to solve the surface integral.

```
P = 2x^3 - y^3
Q = x^3 + y^3
f = d(Q,x) - d(P,y)
x = r cos(theta)
y = r sin(theta)
defint(f r, r, 0, 1, theta, 0, 2 pi)
```

$\frac{3}{2}\pi$

The `defint` integrand is $f\,r$ because $r\,dr\,d\theta = dx\,dy$.

Now let us try computing the line integral side of Green's theorem and see if we get the same result. We need to use the trick of converting sine and cosine to exponentials so that Eigenmath can find the solution.

```
x = cos(t)
y = sin(t)
P = 2x^3 - y^3
Q = x^3 + y^3
f = P d(x,t) + Q d(y,t)
f = expform(f)
defint(f, t, 0, 2 pi)
```

$\frac{3}{2}\pi$

Example 2. Compute both sides of Green's theorem for $F = (1 - y, x)$ over the disk $x^2 + y^2 \leq 4$.

First compute the line integral along the boundary of the disk. Note that the radius of the disk is 2.

```
-- Line integral
P = 1 - y
Q = x
x = 2 cos(t)
y = 2 sin(t)
defint(P d(x,t) + Q d(y,t),t,0,2pi)
```

8π

```
-- Surface integral
x = quote(x) -- clear x
y = quote(y) -- clear y
h = sqrt(4 - x^2)
defint(d(Q,x) - d(P,y), y, -h, h, x, -2, 2)
```

8π

```
-- use polar coordinates
P = 1 - y
Q = x
f = d(Q,x) - d(P,y) -- do before change of coordinates
x = r cos(theta)
y = r sin(theta)
defint(f r, r, 0, 2, theta, 0, 2 pi)
```

8π

```
defint(f r, theta, 0, 2 pi, r, 0, 2) -- integrate over theta first
```

8π

In this case, Eigenmath solved both forms of the polar integral. However, in cases where Eigenmath fails to solve a double integral, try changing the order of integration.